# DECUS
## PROGRAM LIBRARY

| | |
|---|---|
| DECUS NO. | 8-265 |
| TITLE | TELETYPE PARITY CONVERSION PROGRAM |
| AUTHOR | R. Lee |
| COMPANY | University of Kent at Canterbury<br>Canterbury, Kent, England |
| DATE | June 15, 1970 |
| SOURCE LANGUAGE | PAL III |

# TELETYPE PARITY CONVERSION PROGRAM

DECUS Program Library Write-up

DECUS No. 8-265

## 1. Abstract

This tape contains two programs. The first will convert a symbolic tape in ASCII code with parity into one in ASCII code without parity. The second converts a tape in ASCII code without parity into one in ASCII code with parity. Using these programs on line while typing a symbolic tape will cause a teletype without parity to punch a tape in parity code or vice versa. The programs may easily be converted for use with a high speed reader and punch.

## 2. Requirements

### 2.1 Hardware

The minimum hardware required is a 4K PDP-8, 8/S, 8/I or 8/L with an ASR-33 teletype.

### 2.2 Storage

The programs occupy locations 200 to 267.

## 3. Usage

Load the programs using the BIN loader. For conversion from parity to non-parity, load address 200; for conversion from non-parity to parity load address 201. Turn on LSP. Place tape to be converted in LSR, turn on reader and depress Start. The tape is then read in and the converted tape is punched.

To type program tapes on line do not turn on the reader; type the program using the teletype keyboard.

## 4. Method

4.1 To convert tapes with parity into tapes without parity each character is read in and the 200 channel bit removed if present. The 200 channel bit is then added to every character, except a blank; and the resulting character is punched. Blanks are punched unchanged.

4.2 To convert tapes without parity into tapes with parity each character is read in and the 200 channel bit removed. The character is then inspected by rotating each of the 8 bits into the link in turn. The location "Count2" is incremented each time a one is in the link. "Count2" is then tested and if it is even the character is punched; if it is odd the 200 channel bit is added before punching. Blanks are punched unchanged.

5. Program Listing

```
                /TELETYPE PARITY CONVERSION PROGRAMS
                /START AT 200 FOR PARITY TO NON-PARITY
                /START AT 221 FOR NON-PARITY TO PARITY
                /J.L. 15.6.74
                *200
0200  5202               JMP START1
0201  5223               JMP START2
                /
0202  7200    START1,    CLA
0203  6032               KCC
0204  6046               TLS
0205  6031    READ1,     KSF
0206  5205               JMP .-1
0207  6036               KRB                 /READ CHARACTER
0210  7450               SNA                 /IS IT A BLANK?
0211  5214               JMP PRINT1          /YES,
0212  0221               AND P177            /NO,REMOVE 200 CHANNEL IF ANY
0213  1222               TAD P200            /ADD 200 CHANNEL
0214  6041    PRINT1,    TSF
0215  5214               JMP .-1
0216  6046               TLS                 /PRINT CHARACTER
0217  7200               CLA
0220  5205               JMP READ1
0221  0177    P177,      177
0222  0200    P200,      200
                /
0223  7200    START2,    CLA
0224  6032               KCC
0225  6046               TLS
0226  3265    READ2,     DCA TEMP
0227  6031               KSF
0230  5227               JMP .-1
0231  6036               KRB                 /READ CHARACTER
0232  7452               SNA                 /IS CHARACTER A BLANK?
0233  5256               JMP PRINT2          /YES,
0234  0221               AND P177            /NO,REMOVE 200 CHANNEL
0235  3265               DCA TEMP            /STORE CHARACTER.
0236  1264               TAD M8
0237  3266               DCA COUNT1          /PUT -8 IN COUNT1
0240  3267               DCA COUNT2          /CLEAR COUNT2
0241  1265               TAD TEMP
```

```
0242  7100  AGAIN,  CLL
0243  7010          RAR
0244  7430          SZL             /IS LINK SET?
0245  2267          ISZ COUNT2      /YES,
0246  2266          ISZ COUNT1      /NO,
0247  5242          JMP AGAIN       /MORE BITS TO TEST
0250  1267          TAD COUNT2      /FINISHED
0251  7100          CLL
0252  7010          RAR
0253  7620          SNL CLA         /IS NO. OF HOLES EVEN?
0254  5256          JMP PRINT2      /YES,
0255  1222          TAD P200        /NO,
0256  1265  PRINT2, TAD TEMP
0257  6041          TSF
0260  5257          JMP .-1
0261  6046          TLS             /PRINT CHARACTER
0262  7200          CLA
0263  5226          JMP READ2
0264  7770  M8,     -8
0265  0000  TEMP,   0
0266  0000  COUNT1, 0
0267  0000  COUNT2, 0
AGAIN   0242
COUNT1  0266
COUNT2  0267
M8      0264
PRINT1  0214
PRINT2  0256
P177    0221
P200    0222
READ1   0205
READ2   0226
START1  0202
START2  0223
TEMP    0265
```